
Analyse, simulations d’un amplificateur en classe D 
 

 

J’invite le lecteur à consulter le site pour des informations complémentaires.  
 

Page d’accueil du site Internet :    d’autres pdf, sur différents sujets : 
page d'accueil                    liste des PDF  

 

 

Le but de ce travail est de vérifier par des simulations Pspice, le fonctionnement d’un 

amplificateur en classe D. Le principe de base de ce type d’amplificateur est largement 

diffusé sur le net, et est supposé connu par le lecteur. L’originalité de notre montage est 

d’incorporer le driver des transistors MOS dans la simulation. Néanmoins, on est à la 

limite du nombre de nœuds maximum (75) imposé par la version d’évaluation de Pspice. 

 

1) Introduction  

 

Le domaine d’application des amplificateurs en classe D est l’audio, où le récepteur est un haut-parleur. 

Après celui des amplificateurs de salon et du home cinéma, le marché s’est développé vers les téléphones 

portables, là où on cherche à réduire la consommation, pour gagner en autonomie de batterie : en effet, 

l’amplificateur classe D est à très haut rendement. La structure du schéma analysé ici est de « première 

génération » : PWM pleine onde, puissance nominale de 2,5 W, sortie sur haut-parleur de 8 Ω. Les 

simulations permettent de fouiller un peu le fonctionnement. 

 

2) Description  

 

a) Schéma de principe. 
 

La structure proposée est classique :  

- un PWM (ou MLI) à fréquence constante, 

- mise en forme pour commander des interrupteurs, 

- ces interrupteurs, en agissant en tout ou rien, délivrent en 

sortie un signal prenant 2 valeurs possibles : -E , + E, 

- le récepteur récupère le signal, au travers un filtre passe-bas. 
 

Schéma de principe. 

 

b) Schéma simulé 
 

 
Schéma électronique.  

L’entrée est Vin, la sortie du PWM est c, dupliqué en D, la sortie audio est Out 

 

PMOS : 

IRF9140     

 

 

 

NMOS : 

IRF150 

 

 

Rpull : 330 Ω     Rgp = Rgn = 2,2 Ω     LF = 47 µH     CF = 1 µF     CO = 1000 µF     RL = 8 Ω 

Signal d’entrée Vin : source Pspice SIN    Signal rampe Vr : source Pspice PULSE  
 

Dans cette configuration du LM5110-1, les sorties OUT_A et OUT_B sont identiques. Sur le plan 

fonctionnel, c’est donc équivalent à avoir les grilles reliées entre elles. Dans ce cas, l’étage de sortie de 

puissance, formé d’un PMOS et d’un NMOS, a le même schéma qu’un inverseur CMOS. 

http://serge.dusausay.free.fr/
http://serge.dusausay.free.fr/ss/plan2.html

c) Description détaillée 
 

- Signal d'entrée Vin : est bipolaire de dynamique maximum – 6 V à + 6 V. Ces valeurs sont arbitraires. 

C’est une sortie de pré ampli. La fréquence est de 1 kHz. Cette valeur est représentative d’un signal audio, 

dont le spectre est usuellement [20 Hz ; 20 kHz]. 
 

 

- Le signal rampe Vr, ou dent de scie, est bipolaire, de – 6 V à + 6 V. Ces niveaux doivent encadrer les 

valeurs min et max du signal d’entrée Vin, pour assurer la modulation PWM.  

La fréquence est fixe = 200 kHz, valeur conforme à celles rencontrées dans les amplis classe D, 

usuellement de 100 kHz à 400 kHz. Ce signal s’appelle aussi « porteuse à dent de scie ». 
 

- PWM : le signal d’entrée est comparé à la rampe. Le signal de sortie a son 

rapport cyclique α lié à la valeur de la tension d’entrée. Le comparateur tout ou 

rien est ici un LM111, alimenté en ± 12 V. 
 

Il dispose de 2 sorties, symbolisées ici par c et e, respectivement collecteur et 

émetteur « ouverts », c’est-à-dire connectables par l’utilisateur. La sortie « e » est à 

la masse. On exploite la sortie « c »: son niveau bas est voisin de 0 V (c’est un 

VCEsat du transistor interne de sortie), et son niveau haut égal 5 V (grâce au pull up 

sur VP = 5 V).  

 
L’intérieur  

du LM111 

 

 

- Le signal issu du LM111 est donc de niveau 0 V ; 5 V. Sur le plan théorique, cela peut commander des 

transistors idéaux. Mais dans le cas de transistors de puissance, ce signal est très dégradé par les 

condensateurs d’entrée que présentent les transistors réels. 
 

 

La mise en forme pour commander les 

commutateurs est faite par un LM5110. Ce circuit, 

fabriqué par Texas Instrument, est dédié à la 

commande des MOS de puissance, en commutation. 

Le modèle Pspice est donné sous : 

 http://www.ti.com/product/LM5110/toolssoftware 
 

Il s’agit en fait d’un macro modèle, c’est-à-dire un 

schéma fait de composants discrets associés à des 

sources de tension, sources de courant.  
 

 

Le LM5110 est alimenté ici en 0 V + 12 V (nœud 

nommé VLM dans notre schéma). 

Le signal de sortie attendu (OUT_A, OUT_B) peut 

prendre 2 valeurs possibles : 0 V et + 12 V.  
 

 

- Les résistances de grille limitent les pointes de courant à une valeur compatible admise par le LM5110 

(5A sink / 3A source). Ces pointes de courant sont dues aux charges / décharges des capacités d’entrée 

des MOS de puissance, lors des fronts. Dans ce schéma, on n’a pas cherché à optimiser les valeurs en 

fonction du type (N, P) des transistors. 
 

 

Les IRF9140, IRF150 sont les MOS de puissance disponibles dans eval.lib. Ils sont en cohérence avec le 

cahier des charges : ils doivent commuter à 200 kHz sans dégrader le signal (temps de montée, temps de 

descente bien plus bref qu’une fraction de T = 5 µs), et ne présenter aucune chute de tension à l’état 

passant (rdson bien inférieure à RL = 8 Ω), avec des signaux de commande 0 V ; 12 V. On peut estimer 

qu’ils deviennent « ON » quand VGS > VT pour le NMOS et VGS > VT pour le PMOS. On a 

respectivement VT ≈ 2,8 V, etVT ≈ 3,7 V. En MOS de puissance, il y a plus récent et plus performant, 

mais cela ne change pas le principe de fonctionnement que nous allons voir. 

http://www.ti.com/product/LM5110/toolssoftware

- Ces transistors ont les grilles au même potentiel : 

dans notre application, quand un est conducteur, 

l’autre non. Ils sont connectés sur une alimentation 

de puissance connectée en VDD = 12 V et GND. 
 

L’analyse détaillée des transitions montre 

néanmoins une conduction simultanée des 

transistors ce qui provoque des pics de court-circuit 

sur l’alimentation. C’est le principal défaut de 

cette structure à un seul signal de commande. 
 

Le chronogramme de principe ci-contre illustre ce 

phénomène. 

PMOS : conducteur pour VGS > VT , seuil de 

conduction du PMOS = 3,5 V, soit pour VG < 8,5 V. 

 
NMOS : conducteur pour VGS > VT, seuil de conduction 

du NMOS = 3,5 V, soit pour VG > 3,5 V. 

I(VDD) est idéalisé ici par des pics triangulaires. 

On ne tient pas compte des capacités CGS, CGD, qui, 

associées aux résistances internes des transistors, créent 

une déformation et un décalage temporel sur le courant. 
 

Pour ne pas avoir ce défaut de recouvrement, il faudrait modifier les signaux de commande, de façon à 

imposer un temps mort entre les conductions des transistors (commande « break before make »). Cela 

signifie qu’il faut générer 2 signaux, l’un attaquant IN_A, l’autre IN_B, de la forme ci-dessous : 

 
Des signaux de commande présentant un temps mort (tm) pour améliorer la commutation 

 

Cette solution, exploitée en pratique par des circuits spécialisés, entraine un schéma qui présente un 

nombre de nœuds supérieur à 75 et n’est pas simulable avec la version d’évaluation de Pspice. Nous nous 

contenterons de travailler avec une version « lite » du circuit de commande des transistors.  
 

- Le filtre passe bas doit laisser passer la bande audio, et éliminer les harmoniques apportées par la 

modulation de largeur d’impulsion. Il ne doit pas présenter de résistance, qui consommerait de la 

puissance, au détriment du rendement global. Il est proposé ici un basique LC, donc d’ordre 2.  

Pour une gamme de fréquence telle que CO équivalent à un court-circuit, on a :  

VOut(p)/VD(p) 
1(L/R)pLCp

1
2 

  On pose classiquement :  0 = 1/√LC et  z = (0/2)(L/R)  

La fréquence propre est de 1/(2π√LC) = 1/(2π√(47 10-6 1 10-6)) = 23,2 kHz. 

Le coefficient d’amortissement réduit z = 0,429. 
 

- L’alimentation de puissance VDD est 0 V 12 V. Elle est considérée comme parfaite. On n’a pas ajouté 

d’impédance interne, ce qui, en pratique, modifie le courant débité. 
 

- Le potentiel de Drain est donc de part et d’autre de 6 V. Il faut donc couper la composante continue 

avant d’accéder à la charge, ici un haut-parleur, modélisé simplement par une résistance de 8 Ω. Le 

condensateur CO = 1 mF assure ce rôle. Il présente une réactance dont le module vaut 0,16 Ω à 1 kHz : 

on le considère comme un court-circuit devant 8 Ω. 

 

Remarques :  
 

- les alimentations VLM, VDD et Vplus sont égales à 12 V. En simulation, on peut placer une seule 

source parfaite. Il a été choisi de placer 3 sources indépendantes, ce qui laisse la possibilité de modifier 

éventuellement ces tensions, par simple changement de valeur dans la netlist. 
 

- Le point froid est la masse GND, supposée parfaite pour tout le circuit. On ne cherche pas ici à analyser 

les éventuels aléas qui se présentent en pratique quand les signaux analogiques et logiques cohabitent et 

quand on a sur la même carte des signaux basse puissance et forte puissance, ce qui oblige des connexions 

très pertinentes des points de masse pour réduire les bruits de commutation. 

3) Simulations  

 

a) Réponse harmonique du filtre LC seul. 
 

- Dans la netliste, on valide les 2 lignes concernées et on obtient la réponse harmonique : 

           Frequency

10Hz 100Hz 1.0KHz 10KHz 100KHz 1.0MHz

DB(V(Output))

-70

-60

-50

-40

-30

-20

-10

0

 
 

Interprétation :  
 

On obtient bien un plateau 0 dB. Par intersection des asymptotes on peut identifier f0 = 23,2 kHz, ce qui 

est la valeur calculée. On peut lire cette valeur plus précisément par le passage à – 90° de la phase. La 

résonance que l’on voit confirme z = 0,429. 

On mesure une bande passante à – 3 dB : [ 20 Hz ; 31,3 kHz ]. La bande audio est couverte. 

La coupure basse est liée à CO qui ne se comporte plus comme un court-circuit en BF.  

Remarque : à 200 kHz, le filtre a pour module 1/(200k/23,2k)2 = 1/74,3, soit – 37 dB sur la courbe. 

 

b) Réponse temporelle 1 : vérification du fonctionnement de chaque partie 
 

On place ici Vin = 0 V afin de contrôler les différents signaux stratégiques : fonctionnement du PWM, 

commutation des transistors, action du filtre. 

A mi dynamique, le signal PWM sera de rapport cyclique α = 50 %, soit un signal carré de fréquence 

200 kHz. En sortie des transistors, on attend ce signal carré, de niveau théorique 0 V ; 12 V. 

La simulation donne :  

 
 

           Time

39.000ms 39.002ms 39.004ms 39.006ms 39.008ms 39.010ms 39.012ms 39.014ms 39.016ms 39.018ms 39.020ms

V(D)

0V

2V

4V

6V

8V

10V

12V

 
A Vin = 0 V constant, on a en D un signal carré, 0 V ; 12 V 

 

La sortie montre un signal de niveau bas et haut très proche de 0 V et 12 V respectivement, comme 

attendu. Le rapport cyclique α est quasiment 0,5. On profite de ce chronogramme pour mesurer un temps 

de montée (entre 10 % et 90 % par un zoom) de 48 ns et de descente de 12,8 ns. 

On visualise également les courants Grille et courant débité par l’alimentation VDD : 

           Time

39.0000ms 39.0005ms 39.0010ms 39.0015ms 39.0020ms 39.0025ms 39.0030ms 39.0035ms 39.0040ms 39.0045ms 39.0050ms 39.0055ms

- I(VDD1)

0A

10A

20A

30A

40A

50A

IG(MN)

-500mA

0A

500mA

IG(MP)

0A

2.0A

-2.5A

SEL>>

 
On remarque en IG, des pics lors des fronts de la tension de commande VG. On ne dépasse pas la valeur 

admissible du LM5110. 
 

Le courant débité par l’alimentation de puissance VDD montre essentiellement un pic de 46 A par période. 

Cette valeur peut paraitre énorme.  Mais : 

- On a simulé avec l’alimentation VDD parfaite. Une 

impédance interne limiterait ce courant. 

- Si on idéalise grossièrement ce pic à un triangle de 

base 66 ns et de hauteur 46 A, ramené à une période 

T = 5 µs, cela correspond à une valeur moyenne de : 

T

1

2

hauteur x base
 = 

6-

-9

10 5

1

2

 46 x 10 66
 ≈ 0,3 A.            Time

39.00000ms 39.00004ms 39.00008ms 39.00012ms 39.00016ms 39.00020ms 39.00024ms 39.00028ms

- I(VDD1)

0A

10A

20A

30A

40A

50A

 
Zoom sur l’appel de courant 

 

Remarque : la convention de signe utilisée par Pspice justifie la présence du signe « - » dans l’affichage 

de courant d’alimentation. En effet, la convention est « quadripole », ou, dit autrement, « égoïste » (ce qui 

sort est négatif…). Donc le courant d’alimentation VDD qui se dirige vers les transistors est négatif. 

 

Faisons le bilan des puissances, par 

intégration sur quelques périodes, soit 40 µs, 

et ce en régime stabilisé, vers 40 ms : 

 

Remarque : pour ne présenter que des 

grandeurs positives, on a inversé le signe de 

la puissance fournie par VDD. 
           Time

39.00ms 39.01ms 39.02ms 39.03ms 39.04ms 39.05ms 39.06ms 39.07ms 39.08ms 39.09ms 39.10ms

-AVGX(W(VDD1),40u) AVGX(W(MN),40u) AVGX(W(MP),40u) AVGX(V(Gn)*IG(MN),40u) AVGX((V(Gp)-V(VDD))*IG(MP),40u)

0

4.0

 
Puissances mises en œuvre. 

 

L’alimentation fournit 3,6525 W. 

Le NMOS dissipe 1,888 W, dont 90,6 mW apporté par le driver. 

le PMOS dissipe 1,8976 W, dont 37 mW apporté par le driver. 

Les transistors consomment sur VDD :  

(1888 - 90,6) + (1898 - 37) ≈ 1797 + 1861 = 3,658W.  
On retrouve, aux arrondis et fluctuations près, la puissance fournie par l’alimentation VDD : 3,6525 W. 

D’ailleurs on avait prédéterminé cette valeur, par P = VDD Imoy = 12 x 0,3 = 3,6 W. 

Dans cet essai, la charge RL ne dissipe aucune puissance, puisque la valeur moyenne de la tension en D 

est quasi nulle. 

 

Cet essai montre la conséquence néfaste de ce choix de commande identique sur les 2 grilles des 

transistors : sans aucune puissance transmise, on a déjà environ 3,7 W dissipée dans les transistors. 

Vérifions le rôle du filtre, en affichant les spectres en tension : 
 

           Frequency

0Hz 0.4MHz 0.8MHz 1.2MHz 1.6MHz 2.0MHz 2.4MHz 2.8MHz 3.2MHz 3.6MHz 4.0MHz 4.4MHz 4.8MHz

V(D)

0V

2.0V

4.0V

6.0V

8.0V

 
Spectre en D, tension du PWM à α = 0,5 

 

Interprétation (consulter l’annexe A) :  
 

La raie 0 est la valeur moyenne (idéalement 6 V car α = 0,5).  

Puis on retrouve les raies aux fréquences 200 kHz, 600 kHz, 1 MHz…. L’amplitude du fondamental est à 

quasiment à la valeur théorique 7,64 V.  
 

Observons le rôle du filtre, par affichage fréquentiel et temporel de la sortie Out : 

           Frequency

0Hz 0.4MHz 0.8MHz 1.2MHz 1.6MHz 2.0MHz 2.4MHz 2.8MHz 3.2MHz 3.6MHz 4.0MHz 4.4MHz 4.8MHz

V(Out)

0V

20mV

40mV

60mV

80mV

100mV

120mV

    
           Time

39.050ms 39.051ms 39.052ms 39.053ms 39.054ms 39.055ms 39.056ms 39.057ms 39.058ms 39.059ms 39.060ms

V(Out)

-100mV

0V

100mV

 
 

spectral et temporel en Out avec α = 1/2 (Vin = 0 V) : environ 100 mV d’amplitude à 200 kHz 

 

Interprétation :  
 

On a V(Out) =  0 V (la valeur moyenne de 6 V a été coupée par CO)  

+ du bruit additif haute fréquence. 

Ce dernier peut se calculer, d’après ce qui est indiqué en annexe A:  

L’amplitude du fondamental = 7,64 V. A 200 kHz, le filtre atténue d’un coefficient 74,3 : la composante à 

200 kHz vaut 7,64 / 74,3 = 102 mV d’amplitude. La composante à 600 kHz est négligeable et est à peine 

visible sur l’affichage. Cela explique la mesure de  ± 100 mV càc en résiduel sur V(Out). 

 

c) Réponse temporelle 2 : autre point de fonctionnement à Vin constant 
 

On place Vin = - 4,5 V, ce qui représente 1,5 V au-dessus du minimum, soit un rapport cyclique 

théorique de α = 0,125 (obtenu par 1,5/12), la fréquence porteuse restant à 200 kHz.  
 

           Time

39.000ms 39.002ms 39.004ms 39.006ms 39.008ms 39.010ms 39.012ms 39.014ms 39.016ms 39.018ms 39.020ms

V(D)

0V

2V

4V

6V

8V

10V

12V

   
           Frequency

0Hz 0.5MHz 1.0MHz 1.5MHz 2.0MHz 2.5MHz 3.0MHz 3.5MHz 4.0MHz 4.5MHz 5.0MHz

V(D)

0V

0.5V

1.0V

1.5V

2.0V

2.5V

3.0V

 
Temporel et spectral de la sortie D avec α = 1/8 (Vin = -4,5 V) 

Interprétation :  
 

Temporel : La sortie de puissance est quasi parfaite : le driver et les transistors de puissance sont 

dimensionnés pour notre application. 
 

Fréquentiel :  On retrouve les raies aux multiples de 200 kHz, c’est-à-dire 200, 400, 600 kHz. 

L’enveloppe est en sinus cardinal, dont le premier zéro est à 200 kHz / 0,125 = 1,6 MHz. (On perçoit ici 

une raie de très faible amplitude à 1,6 MHz, car on s’éloigne un peu d’un signal parfait). 

La raie 0 Hz est la valeur moyenne. On est très proche de la théorie, établie avec des signaux parfaits, qui 

donne 0,125 x 12 = 1,5 V. 

 

d) Réponse temporelle 3 : caractérisation de l’amplificateur de classe D. 
 

C’est le fonctionnement nominal : le signal d’entrée est sinusoïdal. 
 

Une commande MLI est alors caractérisée par 2 paramètres :  

 
 

▪ m, indice de modulation = 
Vin de fréquence

Vr de fréquence
. 

Dans notre simulation, on a fixé le signal d’entrée à 1 kHz, on a m = 200 kHz / 1 kHz  = 200. 

 

 

▪ r, taux de modulation = 
Vr de dynamique

Vin de dynamique
 

Si l’amplitude de Vin est très faible, r ≈ 0, α reste voisin de 0,5,  
 

Si l’amplitude de Vin est élevée, on exploite toute la dynamique du rapport cyclique : α évolue 

entre 0 et 1 tandis que Vin évolue entre -3 V et 3 V. On a alors r ≈ 1. Remarquons qu’à α proche de 

0 ou 1, le PWM délivre des impulsions très fines. 

Usuellement, on s’interdit un signal d’entrée qui sortirait de l’intervalle [ -3 V ; + 3 V ], car le 

comparateur serait bloqué durant un certain temps, et il n’y aurait plus de modulation. La 

conséquence serait une augmentation de la distorsion du signal reconstitué par le filtre. 

 

Intuitivement, on a intérêt à avoir α qui balaie de 0 à 1, c’est-à-dire r voisin de 1, pour exploiter au mieux 

la modulation. Nous allons vérifier cela sur 2 runs, en conservant m = 200. 

 

Run à r = 0,1 (amplitude de Vin = 0,6 V donc 5,4/12 < α < 6,6/12, c’est-à-dire de 0,45 à 0,55) 

           Time

37.0ms 37.2ms 37.4ms 37.6ms 37.8ms 38.0ms 38.2ms 38.4ms 38.6ms 38.8ms 39.0ms 39.2ms 39.4ms 39.6ms 39.8ms

V(Out)

-600mV

-400mV

-200mV

-0mV

200mV

400mV

600mV

V(in)

-400mV

-200mV

-0mV

200mV

400mV

600mV

SEL>>

 
  A faible taux de modulation, le signal Out, en sortie de l’amplificateur, est très « bruité » 

           Frequency

0Hz 20KHz 40KHz 60KHz 80KHz 100KHz 120KHz 140KHz 160KHz 180KHz 200KHz 220KHz 240KHz

V(Out)

0V

100mV

200mV

300mV

400mV

500mV

600mV

 
Le signal de sortie Out est composé essentiellement des composantes (mesurées après un zoom) : 

 à 1 kHz : 596 mV (signal dominant) 

à 199 kHz et 201 kHz : ≈ 7 mV (négligeable) 

200 kHz : 99,3 mV (forme le « bruit » visualisé) 

 

Pour expliquer la présence des composantes hautes fréquences qui polluent la sinusoïde à 1 kHz en sortie, 

observons le spectre en D, sortie PWM de puissance : 

           Frequency

0Hz 0.5MHz 1.0MHz 1.5MHz 2.0MHz 2.5MHz 3.0MHz 3.5MHz 4.0MHz 4.5MHz 5.0MHz

V(D)

0V

1.0V

2.0V

3.0V

4.0V

5.0V

6.0V

7.0V

8.0V

 
Le spectre PWM est riche de composantes aux multiples de 200 kHz. 

 

           Frequency

0Hz 1KHz 2KHz 3KHz 4KHz 5KHz 6KHz 7KHz 8KHz 9KHz 10KHz

V(D)

0V

1.0V

2.0V

3.0V

4.0V

5.0V

6.0V

 
Zoom vers 0 Hz :  

le PWM montre une raie 0 Hz de 5,94 V qui sera 

coupée et une raie à 1 kHz d’amplitude 597 mV 

qui traverse le filtre et qui forme le signal audio. 

           Frequency

190KHz 192KHz 194KHz 196KHz 198KHz 200KHz 202KHz 204KHz 206KHz 208KHz 210KHz

V(D)

0V

1.0V

2.0V

3.0V

4.0V

5.0V

6.0V

7.0V

8.0V

 
Zoom vers 200 kHz :  

le PWM montre : 

des raies à 199 kHz et 201 kHz : 584 mV 

et surtout celle à 200 kHz : 7,54 V 
 

Ce spectre en amont de la cellule de filtrage explique donc les composantes du signal audio restitué, en 

aval de cette cellule :  1 kHz à 597 mV d’amplitude (dans la bande passante du filtre) 

et de :   variations au voisinage de 200 kHz d’environ 7,55/74,3 = 101 mV. 

 

Il est clair que la contribution de 101 mV est importante devant 583 mV ! Mais dans le cas d’un signal 

audio, ces harmoniques ne seront pas audibles. 

Run à r = 0,833 (amplitude de Vin = 5 V, donc 1/12 < α < 11/12, c’est-à-dire de 0,0833 à 0,916) 

 

           Time

37.0ms 37.2ms 37.4ms 37.6ms 37.8ms 38.0ms 38.2ms 38.4ms 38.6ms 38.8ms 39.0ms 39.2ms 39.4ms 39.6ms 39.8ms

V(Out)

-5.0V

-4.0V

-3.0V

-2.0V

-1.0V

0V

1.0V

2.0V

3.0V

4.0V

5.0V

V(in)

-5.0V

-3.0V

-2.0V

-1.0V

0V

1.0V

2.0V

3.0V

4.0V

5.0V

SEL>>

 
A fort taux de modulation, le signal Out, en sortie de l’amplificateur, est très faiblement bruité 

 

Le signal d’entrée et celui de sortie sont quasiment identiques. Rappelons que les amplitudes devraient 

être les mêmes (conséquence du choix de ± 6 V pour Vin et Vr, et de ± 6 V pour VD). 

 

On vérifie que la fonction PWM est assurée : 
 

           Time

37.240ms 37.242ms 37.244ms 37.246ms 37.248ms 37.250ms 37.252ms 37.254ms 37.256ms 37.258ms 37.260ms

V(D)

-1V

0V

1V

2V

3V

4V

5V

6V

7V

8V

9V

10V

11V

12V

13V

 
Détail au voisinage de α minimum :  

400 ns à l’état haut, soit α ≈ 0,08 

           Time

37.740ms 37.742ms 37.744ms 37.746ms 37.748ms 37.750ms 37.752ms 37.754ms 37.756ms 37.758ms 37.760ms

V(D)

-1V

0V

1V

2V

3V

4V

5V

6V

7V

8V

9V

10V

11V

12V

13V

 
Détail au voisinage de α maximum :  

450 ns à l’état bas, soit α ≈ 0,9 

L’électronique « suit » (commande et puissance) et peut délivrer ces impulsions fines. 

 

Expliquons également le signal en Out par le spectre en D, sortie du PWM de puissance. 

           Frequency

0Hz 0.5MHz 1.0MHz 1.5MHz 2.0MHz 2.5MHz 3.0MHz 3.5MHz 4.0MHz 4.5MHz 5.0MHz

V(D)

0V

1.0V

2.0V

3.0V

4.0V

5.0V

6.0V

 
Spectre de l’onde PWM de puissance : 

des raies sont regroupées autour des multiples de la fréquence porteuse. 



Comprendre l’électronique par la simulation, par S. Dusausay              article 57                         Page  10 / 13         2024/2025 
 

           Frequency

0Hz 1KHz 2KHz 3KHz 4KHz 5KHz 6KHz 7KHz 8KHz 9KHz 10KHz

V(D)

0V

1.0V

2.0V

3.0V

4.0V

5.0V

6.0V

 
 

Zoom vers 0 Hz :  

le PWM est composé d’une raie 0 Hz de 6 V,  

et d’une raie à 1 kHz d’environ 5 V 

 

           Frequency

190KHz 192KHz 194KHz 196KHz 198KHz 200KHz 202KHz 204KHz 206KHz 208KHz 210KHz

V(D)

0V

1.0V

2.0V

3.0V

4.0V

 
Zoom vers 200 kHz :  

Des raies sont présentes tous les kHz,  

mais l’amplitude max est à 3,5 V 
 

Le signal audio, en sortie du filtre sera donc composée de 1 kHz à 5 V d’amplitude (dans la bande 

passante du  filtre) + des variations au voisinage de 200 kHz proches de 3,5/74,3 = 47 mV. Cela explique 

l’aspect plus pur de la sinusoïde. 

 

e) Bilan des puissances, rendement 
 

Profitons de ce dernier essai, à fort taux de modulation, plus représentatif, pour faire le bilan des 

puissances. 
 

En supposant 5 V d’amplitude, par loi de Joule, la puissance attendue dans RL est (5/√2)2/8 = 1,56 W. 
 

On choisit une intégration sur deux périodes, soit 2 ms. 

Le NMOS dissipe 1,89 W, le PMOS dissipe 1,91 W. 

La charge RL consomme 1,52 W, VDD fournit 5,2 W.  

Si on limite le calcul à la partie puissance, le rendement n’est que : 

1,52 / 5,2 = 30 %.  

Remarquons que si on retire la puissance de court-circuit (établie il est vrai pour Vin constant), l’alim 

fournit 5,2 W - 3,6 W = 1,6 W. Le rendement passe à 95 % ce qui est une valeur plus cohérente. 

 

4) Sans recouvrement, l’amplificateur classe D a un très bon rendement  
 

Dans nos simulations, la puissance de court-circuit (à cause des 

signaux de commande non optimisés) a eu comme conséquence un 

rendement très médiocre pour un classe D. Nous proposons ici un 

artifice de simulation pour représenter le fonctionnement 

comportemental, en remplaçant les transistors IRF par des 

interrupteurs parfaits, disponibles dans la librairie de Pspice. 

Ces interrupteurs sont les composants « Sxxx », qui sont des 

quadripôles : 2 broches où l’on applique la tension de commande, et 

2 broches qui présentent une résistance commandée. 

On a laissé Rgp, Rgn, pourtant devenues inutiles. 

 
Les transistors remplacés 

par des « SWITCH » 
 

En paramétrant judicieusement les 2 composants S,  

on peut assurer un temps mort entre les 2 conductions.  

Pout Sintp :  0,01 Ω pour une tension < 1 V,  

  1 MΩ pour une tension > 2 V. 

Pout Sintn :  0,01 Ω pour une tension > 11 V, 

  1 MΩ pour une tension < 10 V. 

          Sintp : conducteur pour VGP <1 V. 

 
          Sintn : conducteur pour VGN >11 V. 

Si la pente en VGp ou VGn dure 12 ns pour franchir 12 V (ordre de grandeur), on aura 8 ns de temps mort. 

Les switches, opérateurs parfaits, ont un temps de réponse nul. 

Durant ce temps mort, le potentiel de D est flottant (en fait relié à VDD par 1 MΩ, et relié à GND par 

1 MΩ). Il faut placer un condensateur entre le nœud D et la masse, pour laisser circuler le courant 

selfique. Sinon, des pics de tension se présentent au nœud D, à cause de la présence de Ldi/dt. 

Relançons la même simulation avec ces interrupteurs parfaits. 

On ne reproduit ici que le test du rendement : 

           Time

38.0ms 38.2ms 38.4ms 38.6ms 38.8ms 39.0ms 39.2ms 39.4ms 39.6ms 39.8ms 40.0ms

-AVGX(W(VDD1),2m) AVGX(W(RL),2m) AVGX(W(Sintn),2m) AVGX(W(Sintp),2m) AVGX(W(Cam),2m)

0

0.5

1.0

1.5

2.0

 
 

On choisit une intégration sur deux périodes, soit 2 ms. 
 

VDD fournit 1,8484 W, la charge RL consomme 1,5572 W.  
 

Les interrupteurs (qui présentent 0,01Ω / 1 MΩ) dissipent 80 mW. 

On pourrait ramener cette valeur à 0 W en plaçant des valeurs encore 

plus extrêmes.  
 

Le condensateur de 10 nF placé pour éviter les pics de tension consomme de la puissance : 209 mW. 
 

Le bilan est : 1,5572 + 0,08 + 0,2094 = 1,8466, ce qui est parfaitement conforme à 1,8484 W affiché. 
 

Le rendement est 1,5572/1,8484 = 84,2 % . 

 

 

 

5) Conclusion  

 

Le but de ce document était de montrer le principe d’un amplificateur classe D au travers une simulation 

concrète. Le choix de transistors de puissance NMOS, PMOS était dicté par des raisons de simplicité, 

l’étage de puissance se comportant comme un inverseur CMOS, avec un signal de commande équivalent 

unique. Dans ces conditions, malgré un fonctionnement satisfaisant, on a pu mettre en avant le défaut de 

la puissance de court-circuit qui fait chuter le rendement. 

 

Industriellement, on privilégie les NMOS (plus performants) pour les 2 interrupteurs, ce qui modifie le 

signal de commande d’une part, et le circuit de polarisation d’autre part (technique du « bootstrap » pour 

relever une tension). Dans ces conditions, le rendement est très amélioré.  

 

La simulation avec les transistors remplacés par des interrupteurs parfaits a montré une très nette 

amélioration du rendement.  

Annexe A : Série de Fourier du train d’impulsions 
 

Un signal périodique, appelé train d’impulsions, de période T, et de rapport cyclique α, d’amplitude A, 

s’écrit, par décomposition en série de Fourier, en choisissant une bonne origine des temps : 







n

n 1

nn0 tω cos bb ,  où b0 = A α,  et les coefficients bn , pour n ≥ 1, = 
u

usin 
A2α avec u = n π α. 

Exemple 1 :  α = 0,5. Avec A = 12. 

Les coefficients bn sont alors :  
 

rang   n = 1 n = 2 n = 3 n = 4 n = 5 n = 6 … 

u = n π α   π/2 π 3π/2 2π 5π/2 3π  

(sin u)/u   0,636 0 -0,212 0 0,127 0  

bn b0 = 

6 u

usin 
A2α  

b1 = 

7,64 

b2 = 

0 

b3 = 

2,544 

b4 = 

0 

b5 = 

1,528 

b6 = 

0 

 

Nous avons des raies aux fréquences impaires (1, 3, 5… fois le fondamental), avec des amplitudes de type 

1, 1/3, 1/5, 
 

Exemple 2 : α = 0,125. Avec A = 12 : 

Les coefficients bn sont alors :  
 

rang   n = 1 n = 2 n = 3 n = 4 n = 5 … n = 8 

u = n π α   0,125 π 0,25π 0,375π 0,5π 0,625π  π 

(sin u)/u   0,974 0,9 0,784 0,637 0,47  0 

bn b0 = 

1,5 u

usin 
A2α  

b1 = 

2,92 

b2 = 

2,7 

b3 = 

2,35 

b4 = 

1,91 

b5 = 

1,411 

 

 

b8 = 

0 

Nous avons des raies aux fréquences multiples du fondamental, avec des amplitudes enveloppées par un 

sin cardinal. 

 

Annexe B : Spectre d’une onde PWM 
 

Le spectre de l’onde PWM dont la commande fluctue en fonction du temps est complexe.  

En toute rigueur mathématique, si le modulant est périodique, le spectre sera formé de raies. Ce cas n’est 

pas fréquent, sauf le cas d’école : en manipulation avec un GBF délivrant un signal simple, ou le cas de 

notre simulation (où on est, de plus, synchrone). Mais un modulant est généralement aléatoire (par 

exemple un signal issu d’une source audio), mais est à spectre limité. 
 

On montre que selon le taux de modulation r (r est faible si le rapport cyclique évolue peu, r est proche de 

1 si le rapport cyclique varie pleinement), le spectre est différent : 
 

S’il n’y a aucune modulation (r = 0), le PWM délivre un train d’impulsions. Le spectre est donc formé de 

raies placées aux multiples de la fréquence porteuse. Si de plus, on est à Vin au milieu de la dynamique, 

(= 0 V dans notre schéma) on n’a que les multiples impairs, comme indiqué ci-dessus. 
 

S’il y a un peu de modulation, le spectre du 

modulant est présent, mais de faible niveau. 

Il est dupliqué de part et d’autre de fp, 2fp… 

avec un niveau relativement important. 
 

Si la modulation est profonde, le spectre du 

modulant est présent, mais de fort niveau. Il 

est dupliqué de part et d’autre de fp, 2fp… 

avec un niveau relativement faible. 

 

 
fp est la fréquence du PWM, soit 200 kHz. 

Le modulant est le signal audio 
 

Le filtre passe bas doit donc laisser passer la fréquence haute du modulant (en rouge), et couper bien 

avant la fréquence du PWM. Cela explique qu’il est préférable d’avoir fp >> fmodulant (soit m élevé). Et, 

pour que le signal récupéré soit de fort niveau, qu’il est préférable d’avoir r proche de 1. 



Comprendre l’électronique par la simulation, par S. Dusausay              article 57                         Page  13 / 13         2024/2025 

 

 

Annexe C : Netlist 

 
Ampli classe D avec driver LM5110 
*fichier classeD.cir 
 
.lib eval.lib  ; contient LM311 IRF9140 et IRF150 
.lib LM5110_1M_TRANS.lib  ; fournie par TI 
 
.Model intp Vswitch (Ron=0.01 Roff=1Meg Von=1 Voff=2) ; ON à 0 V  
.Model intn Vswitch (Ron=0.01 Roff=1Meg Von=11 Voff=10) ; ON à 12 V 
 

VDD1 VDD 0 DC 12 ; alim puissance  
VDD2 VLM 0 DC 12 ; alim LM  
VDD3 VP 0 DC 5 ; alim pullup pour le LM111 
Vplus Vplus 0 DC 12  ; pour le LM111 
Vmoins Vmoins 0 DC -12 ; pour le LM111 
 
MP D Gp VDD VDD IRF9140 ; PMOS  D G S Sub 
MN D Gn 0 0 IRF150 ; NMOS  D G S Sub 
 
*Sintp VDD D Gp 0 intp  ; modele Pmos 
*Sintn 0 D Gn 0 intn  ; modele Nmos 
*Cam D 0 10n 

 
*Ventree in 0 DC 0 ; pour tester alpha = 50% 
*Ventree in 0 DC -4.5 ; pour tester alpha = 12,5 % 
*Ventree 0 in sin (0 5 1k) ; pour caractériser 
 
Vosc R 0 pulse (-6 6 0 4.98u 10n 10n 5u) ; Rampe 200kHz 
 
X1 R in Vplus Vmoins c 0 LM111 ; comparateur e+ e- V+ V- Sc Se 
Rpull c VP 330 ; pour le collecteur ouvert 

X2 c c VLM 0 VLM outa outb 0 LM5110_1M  
* IN_A IN_B nSHDN IN_REF VCC OUT_A OUT_B VEE 
Rgp outa Gp 2.2 ; pour limiter courant dans LM5510 
Rgn outb Gn 2.2 ; pour limiter courant dans LM5510 
 
Lf D F 47uH  ; filtre  
Cf F 0 1uF ; ordre 2 
Co F Out 1000uF ; coupe DC 
RL Out 0 8 ; Haut parleur 
* constante de temps de 8ms 
 
.IC V(F)=6 V(Out)=0 

 
* 2 lignes à valider pour réponse filtre uniquement : 
VtestAC D 0 AC=1 
.AC DEC 1000 10 1MEG 
* 
* à valider pour fonctionnement 
*.TRAN 1u 35m 30m 1u UIC 
 
.probe 
.end 

Netlist, prête à simuler 

 

Commentaire : 

 

Le run va de 0 à 40 ms. Pour éviter de visualiser le régime transitoire lié à la charge de C0, on 

n’enregistre pas de 0 à 30 ms. 

 

 

Avertissement : 

 

Comme dans tout montage qui exploite des signaux basse et haute fréquences (entrée en BF et porteuse 

en HF), il faut faire un run sur une longue durée (quelques périodes de basse fréquence) et avec une 

définition poussée (pour travailler avec des temps plus court qu’une période de porteuse).  Pour améliorer 

la résolution fréquentielle lors de la FFT, il faut une acquisition de points sur plusieurs périodes de BF. 

 

Tout cela entraine une longue durée de run de simulation, un fichier de résultats volumineux (des 

centaines de MO), et un affichage des courbes très long à venir : celui de la puissance par la commande 

Probe AVGX(,) prend plusieurs minutes, même sur un PC puissant… 

